Introduction to Stata Workshop Notes
Christine Leibbrand, CSDE

*Based off Cori Mar’s Introduction to Stata 2018 Workshop

RESOURCES
NLSY website
https://www.bls.gov/nls/nlsy97.htm

NLS Investigator
https://www.nlsinfo.org/investigator/pages/login.jsp

UCLA IDRE State Resources
https://stats.idre.ucla.edu/stata/

Stata Corporation Training
https://www.stata.com/learn/

Stata Corporation Online (free) Tutorials
https://www.stata.com/links/video-tutorials/

Free Instructional Stata Webinars
https://www.stata.com/training/webinar/

BASICS

Main Windows
1) Results – output and messages about your Stata code (including errors)
2) Command – where you type in your code

Side Windows:
1) Review – a record of the code you’ve typed in your command box and the success/failure of that code
2) Variables – a list of variables in the data memory
3) Properties – information about the selected variable and dataset
*You can adjust the size of these windows by dragging the edges.

Do-File Editor
A do-file editor is a separate, optional window in which you can type your data code. While optional, it is best practice to always code in this window in order to keep a record of your work.

Instructions for opening a Do-File Window. Click on the “Windows” tab on the drop-down menu located at the top of the Stata screen, click on “Do-File Editor” towards the bottom of the drop-down menu and then click “New do-file editor.”

Once this window is open, you can type your code into the window. Save your code by clicking “File > Save as” in the dropdown menu at the top of the .do file. Make sure to pick a meaningful name for your .do file that will help distinguish it from other .do files you might make in the future.

Import Data
Stata datasets have the extension .dta
1) To import a dataset from the dropdown menu, click the “File” tab, then “Open,” then navigate to your dataset, click on it, and then click “Open”

What changes in your Review and Variable windows?

2) Do-File Editor: click once on the Review window’s entry that starts with “use…” From the command window, copy and paste the code that appears into the do-file editor window. Highlight that command line and then click “Execute” in the .do file (the image on the toolbar that looks like a document with an > arrow).

Commenting in Your Do-File
Including Comments in your do-file is very important for documenting what you are doing, both for yourself and for anyone who might need to look at your code (such as coauthors). It also makes it easier to navigate through your .do file when you have a lot of code.

You can include comments in the following ways:
1) Add a * to the beginning of your line of code for one-line comments. Example below:

*Model 1: Bivariate relationship between migration status and income

2) Add /* to the front of the comment, and */ to the end of the comment for multi-line comments. Example below:

/*
191206_IntrotoStataWksp_Tutorial.do
Christine Leibbrand
12/6/2019
Stata Do-File for CSDE Intro to Stata Workshop
*/

Including a header is a best practice for keeping files organized. In general, your header should include:
	Filename
	Author Name
	Date
	Description of what the code does



Get to Know Your Data
Before getting started, it’s helpful to have an idea of what your data looks like. Review the NLSY_Workshop_Codebook in the workshop materials folder to get a sense of NLSY variable names and distributions. 

You can also get to know your data in a variety of ways in Stata. To view information about your data, type the following commands into your command window:

describe

How many observations (obs) are in the dataset?
How many variables (vars) are in the dataset?

codebook

What kind of information does codebook give you?

browse 

To look at the data by observation, use list:

help list
help varlist

list CVC_HIGHEST_DEGREE_EVER_XRND in 1/10
list CVC_HIGHEST_DEGREE_EVER_XRND KEYSEX_1997 in 5000/5005
list CVC_HIGHEST_DEGREE_EVER_XRND KEYSEX_1997 in -5/-1
list CVC_SAT_MATH_SCORE_2007_XRND*
list _all

To look at categorical frequencies, use tab (or tabulate)

tab KEYSEX_1997
tab KEYRACE_ETHNICITY_1997

Feeling lost?

Help
help describe


CUSTOMIZING YOUR DATA: RENAMING AND LABELING

Generally, the variable names and labels that are automatically applied will not make much sense (though the NLSY datasets tend to have reasonably intuitive names and labels). It’s therefore helpful to rename variables and label them appropriately so that we can easily analyze our data and results.

Renaming variables

help rename
rename KEYSEX_1997 gender_1997
rename KEYRACE_ETHNICITY_1997 raceeth_1997
rename YSCH_37000_1997 safesch_1997

Labeling variables

help label
label variable gender_1997 "KEYSEX_1997, self-reported gender"
label variable raceeth_1997 "KEYRACE_ETHNICITY_1997, self-reported race/ethnicity"
label variable safesch_1997 "YSCH_37000_1997, respondent report of agreement they feel safe at school"

tab gender_1997
tab raceeth_1997
tab safesch_1997

EXERCISE 1: Renaming and Labeling Variables

 Rename the following variables and include a descriptive variable label:

Old Name				             New Name	  Variable Label
CV_HGC_BIO_DAD_1997		             dadedu_1997	  Bio dad’s highest grade completed

CV_HGC_BIO_MOM_1997		             momedu_1997 Bio mom’s highest grade completed	

CVC_HIGHEST_DEGREE_EVER_XRND  ownedu_1997   Own highest grade completed

YSAQ_INTRO_3_1997			 favicecr_1997   Favorite ice cream flavor

YHEA_100_1997				 genhealth_1997 Reported level of general health

Labeling the values of a variable
To label the values of a variable, you need to go through two steps: 1) Define your labels and 2) Apply those labels to your variable as shown below:

help label

Example 1:

1) Define label

label define racelabel 1 "1 Black" 2 "2 Hispanic" ///
3 "3 Mixed Race (Non-Hispanic) " 4 "4 Non-Black/Non-Hispanic"

*See note below for multi-line comments

label list

2) Apply those labels to your variable: Currently, our label has not been assigned to any variable

label values raceeth_1997 racelabel

3) Check work

tab raceeth_1997

*label drop racelabel

Note (Multi-Line Comments): Stata treats each line as a separate data command unless “///” is specified at the end of your line of code. In this case, Stata will continue to read the subsequent line of code. This only works in a .do file, however. Stata will not recognize “///” if it is included in the command window.

Example 2:

label define genderlabel 1 "1 Male" 2 "2 Female" 0 "0 No Information" 

label list

label values gender_1997 genderlabel

tab gender_1997

Example 3:

label define genhealthlab -4 "-4 Valid Skip" ///
-3 "-3 Invalid Skip" ///
-2  "-2 Don’t Know" ///
1 "1 Excellent" ///
2 "2 Very Good" ///
3 "3 Good" ///
4 "4 Fair" ///
5 "5 Invalid Skip"

label values genhealth_1997 genhealthlab

tab genhealth_1997

EXERCISE 2: Create and apply a value label to ownedu_1997 and then tabulate to check your work

EXERCISE 3: Create and apply a value label to safesch_1997 and then tabulate to check your work

CLEANING MISSING VALUES

We must set values as missing if we do not want those values included in our analyses. Otherwise they will be treated as real values in regressions and descriptive statistics. 

help missing

To set values as missing, we can use the following strategies: (1) recode or (2) mvdecode

Recode

Recode missing values for each variable by using recode 

help recode

Example 1: Recode a single missing value (a variable has only one value that represents missingness)

1) Tabulate variable to check for missing or unusual values

tab ownedu_1997

2) Set missing values equal to . (Stata treats values starting in “.” As missing)

recode ownedu_1997 (-3 = .)

3) Check your work

tab ownedu_1997
tab ownedu_1997, missing

Example 2: Recode multiple missing values for a variable and label them with their distinct type of missingness (.i for invalid skips, .v for valid skips, .u for unknown)

tab momedu_1997
recode momedu_1997 (-3 = .i) (-4 = .v) (95 = .u)
tab momedu_1997, missing

Example 3: Recode multiple missing values for a variable and do not distinguish distinct types of missingness

recode safesch_1997 genhealth_1997 (-1 -2 -3 -4 = . )

tab safesch_1997, m
tab genhealth_1997, m

mvdecode

You can also label values as missing in batches

mvdecode _all, mv(-1 = .r \ -2 = .d \ -3 = .i  \ -4 = .v \ -5 = .n \ -7 = .m)

mvdecode varlist, mv(-1 = .r \ -2 = .d \ -3 = .i  \ -4 = .v \ -5 = .n \ -7 = .m)

EXERCISE 4: Set missing values for favicecr_1997 and dadedu_1997

CREATING NEW VARIABLES

Sometimes we will need to create new variables based off of the existing variables in our dataset. 

Example 1: Suppose, for example, we want to create a categorical variable for dad's education with the values representing the following:

0 = Less than High School
1 = HS Degree
2 = Some College
3 = College Degree
4 = College+

To do this, we first create a new variable using "gen" or "generate" and set it equal to missing so that we do not accidentally assign incorrect values:

gen dadedu_cat_1997 = .

Now we add in our values using "replace" and using "if" statements to specify how values will be added in:

Terminology: 
Value Operators: 
x < y : x is less than y
x == y : x equals y—note the two equal signs
x > y : x is more than y 
x <= y : x is less than or equal to y
x >= y x is greater than or equal to y
Adding Multiple Conditions: If you want a variable to take a certain value only if it meets two conditions then use the “&” sign; if you want a variable to take a certain value if it meets one condition or a different condition, use the “|” sign

replace dadedu_cat_1997 = 0 if(dadedu_1997 < 12)
replace dadedu_cat_1997 = 1 if(dadedu_1997 == 12)
replace dadedu_cat_1997 = 2 if(dadedu_1997 > 12 & dadedu_1997 < 16)
replace dadedu_cat_1997 = 3 if(dadedu_1997 == 16)
replace dadedu_cat_1997 = 4 if(dadedu_1997 > 16 & dadedu_1997 < 95)

Example 2: Create a dichotomous variable indicating if the respondents agrees or strongly agrees that they feel safe at school (= 1) or if the respondent disagrees or strongly disagrees that they feel safe at school (= 0) based on the responses to the question safesch_1997.

gen safesch_cat_1997 = .
replace safesch_cat_1997 = 1 if(safesch_1997 == 1| safesch_1997 == 2)
replace safesch_cat_1997 = 0 if(safesch_1997 >= 3 & safesch_1997 <= 4)

EXERCISE 5: Create a dichotomous variable indicating if the respondent reports having excellent/very good/good health (=1) or whether the respondent reports having fair/poor health (=0) in response to the genhealth_1997 measure.

SUBSETTING YOUR DATA

Subsetting Variables
Sometimes we will not need all of the variables included in our dataset. For example, we may want to just analyze the variables we have labeled and we do not need any of the other variables included in the data. Removing these unwanted variables is very simple. However, always make sure to keep your unique identifier variables! It is easy to forget about these and accidentally delete them, making your data uninterpretable.

keep PUBID_1997 gender_1997 raceeth_1997 momedu_1997 dadedu_1997 ownedu_1997 genhealth_1997 safesch_1997

describe

Alternative strategy: You can right click on variables in the Variable window and delete them. This is not a preferred strategy though as it is best to document which variables you have deleted in your .do file

It may be valuable to save your data with a new name in case you later want to refer back to the larger dataset.

save newfilename.dta, replace

Subsetting Cases

In some cases, we will want to examine a subset of cases in our data, such as only females. To do this, we can use the strategy below:

keep if gender_1997 == 2

describe

save newfilename_female.dta, replace
[bookmark: _GoBack]
